AVR108: Setup and Use of the LPM Instruction

8-bit AVR® Microcontroller

Application Note

Features

- **Use of the LPM (Load Program Memory) Instruction with the AVR® Assembler**
- **Load Constants from Program Memory**
- **Use of Lookup Tables**

Introduction

This application note describes how to access constants saved in Flash Program memory of the AVR microcontrollers. The AVR is based on a Harward architecture, this means that Address and Data memory use separate busses. This is necessary to achieve single cycle instructions execution speed. To be able to save constants in Flash memory the Load Program Memory (LPM) instruction is included in the instruction set.

Use of the LPM

The LPM instruction is included in the AVR Instruction Set to load a data byte from the FLASH Program memory into the Register File.

The Flash Program memory of the AVR microcontroller is organized as 16 bits words. The Register File and SRAM Data memory are organized as eight bits bytes. Special consideration must therefore be taken when loading data from Program memory to Data meory.

The Z-register in the Register File is used to access the Program memory. This 16 bits register pair is used as a 16 bits pointer to the Program memory. The 15 most significant bits selects the word address in Program memory. Because of this, the word address is multiplied by two before it is put in the Z-register.

Figure 1. Z Address Register



The least significant bit of the Z Address Register selects either Low byte (0) or High byte(1) of the Program memory word. To calculate the Low (ZL) an High (ZH) part of the address, use the LOW() and HIGH() functions.

To load data from random places in program memory, the Z-register must be set up with the proper address each time a new address is accessed.

In Program memory the data is organized with one byte in the low part of a program word and the next byte in the high part. Because of this, the message string will appear as if every pair of characters has been swapped, when viewed in the memory view in AVR Studio®.

Rev. 1233B–AVR–05/02

The program in this application note loads a string of bytes from the Program memory, and writes it to Port B. It first initializes Port B so that all the pins are output. It loads the starting address of the string “Hello World” into Z-register, as described above. Then a byte is loaded from program memory using LPM. The program checks whether or not the end of the string is reached (byte was zero). If the end is not reached yet the last read byte is put on Port B, a short delay is made, and the Z-register is increased. The program then jumps back to load another byte.

```
;**** A P P L I C A T I O N   N O T E   A V R 1 0 8 ************************
;*
;* Title:             Load Program Memory
;* Version:           1.0
;* Last updated:      98.12.17
;* Target:            AT90Sxx1x and higher (Devices with SRAM)
;*
;* Support E-mail: avr@atmel.com
;*
;* DESCRIPTION
;* This Application note shows how to use the Load Program Memory (LPM)
;* instruction. The App. note loads the string "Hello World" from
;* program memory byte by byte, and puts it onto port B.
;*
;*******************************************************************************


  .include "8515def.inc"
  .device   AT90S8515          ; Specify device
  .def      temp=r16           ; Define temporary variable


  start:
    ldi     temp,low(RAMEND)
    out     SPL,temp           ; Set stack pointer to last internal RAM
location
    ldi     temp,high(RAMEND)
    out     SPH,temp

    ldi     temp,$ff
    out     PORTB,temp         ; Set all pins at port B high
    out     DDRB,temp          ; Set port B as output

  ; Load the address of 'message' into the Z register. Multiplies
  ; word address with 2 to achieve the byte address, and uses the
  ; functions high() and low() to calculate high and low address byte.

    ldi     ZH,high(2*message) ; Load high part of byte address into ZH
    ldi     ZL,low(2*message)  ; Load low part of byte address into ZL

  loadbyte:
    lpm                        ; Load byte from program memory into r0
```

```
  tst     r0                ; Check if we've reached the end of the message
  breq    quit              ; If so, quit

  out     PORTB,r0          ; Put the character onto Port B
  rcall   one_sec_delay     ; A short delay

  adiw    ZL,1              ; Increase Z registers
  rjmp    loadbyte

  quit:   rjmp quit

one_sec_delay:
  ldi     r20, 20
  ldi     r21, 255
  ldi     r22, 255
delay:
  dec     r22
  brne    delay
  dec     r21
  brne    delay
  dec     r20
  brne    delay
  ret

message:
.db "Hello World"
.db 0
```

Atmel Headquarters

Corporate Headquarters
2325 Orchard Parkway
San Jose, CA 95131
TEL 1(408) 441-0311
FAX 1(408) 487-2600

Europe
Atmel Sarl
Route des Arsenaux 41
Case Postale 80
CH-1705 Fribourg
Switzerland
TEL (41) 26-426-5555
FAX (41) 26-426-5500

Asia
Room 1219
Chinachem Golden Plaza
77 Mody Road Tsimhatsui
East Kowloon
Hong Kong
TEL (852) 2721-9778
FAX (852) 2722-1369

Japan
9F, Tonetsu Shinkawa Bldg.
1-24-8 Shinkawa
Chuo-ku, Tokyo 104-0033
Japan
TEL (81) 3-3523-3551
FAX (81) 3-3523-7581

Atmel Operations

Memory
2325 Orchard Parkway
San Jose, CA 95131
TEL 1(408) 441-0311
FAX 1(408) 436-4314

Microcontrollers
2325 Orchard Parkway
San Jose, CA 95131
TEL 1(408) 441-0311
FAX 1(408) 436-4314

La Chantrerie
BP 70602
44306 Nantes Cedex 3, France
TEL (33) 2-40-18-18-18
FAX (33) 2-40-18-19-60

ASIC/ASSP/Smart Cards
Zone Industrielle
13106 Rousset Cedex, France
TEL (33) 4-42-53-60-00
FAX (33) 4-42-53-60-01

1150 East Cheyenne Mtn. Blvd.
Colorado Springs, CO 80906
TEL 1(719) 576-3300
FAX 1(719) 540-1759

Scottish Enterprise Technology Park
Maxwell Building
East Kilbride G75 0QR, Scotland
TEL (44) 1355-803-000
FAX (44) 1355-242-743

RF/Automotive
Theresienstrasse 2
Postfach 3535
74025 Heilbronn, Germany
TEL (49) 71-31-67-0
FAX (49) 71-31-67-2340

1150 East Cheyenne Mtn. Blvd.
Colorado Springs, CO 80906
TEL 1(719) 576-3300
FAX 1(719) 540-1759

Biometrics/Imaging/Hi-Rel MPU/
High Speed Converters/RF Datacom
Avenue de Rochepleine
BP 123
38521 Saint-Egreve Cedex, France
TEL (33) 4-76-58-30-00
FAX (33) 4-76-58-34-80

e-mail
literature@atmel.com

Web Site
http://www.atmel.com

1233B–AVR–05/02 0M